

# ACANFD\_FeatherM4CAN Arduino library, for Adafruit Feather M4 CAN Version 1.0.0

Pierre Molinaro

March 8, 2022

## Contents

<b>1</b>	<b>Versions</b>	<b>3</b>
<b>2</b>	<b>Features</b>	<b>3</b>
<b>3</b>	<b>CAN Interfaces</b>	<b>3</b>
3.1	CAN0 . . . . .	3
3.2	CAN1 . . . . .	4
<b>4</b>	<b>Data flow</b>	<b>5</b>
<b>5</b>	<b>A simple example: LoopBackDemoCANFD_CAN1</b>	<b>6</b>
<b>6</b>	<b>The CANMessage class</b>	<b>9</b>
<b>7</b>	<b>The CANFDMessage class</b>	<b>9</b>
7.1	Properties . . . . .	10
7.2	The default constructor . . . . .	10
7.3	Constructor from CANMessage . . . . .	10
7.4	The type property . . . . .	11
7.5	The len property . . . . .	12
7.6	The idx property . . . . .	12
7.7	The pad method . . . . .	12
7.8	The isValid method . . . . .	12
<b>8</b>	<b>Transmit FIFO</b>	<b>13</b>
8.1	The driverTransmitFIFOSize method . . . . .	13
8.2	The driverTransmitFIFOCOUNT method . . . . .	13

## CONTENTS

---

8.3	The <code>driverTransmitFIFOPeakCount</code> method . . . . .	13
<b>9</b>	<b>Transmit buffers (<code>TxBuffers<sub>i</sub></code>)</b>	<b>14</b>
<b>10</b>	<b>Receive FIFOs</b>	<b>14</b>
<b>11</b>	<b>Payload size</b>	<b>14</b>
11.1	The <code>ACANFD_FeatherM4CAN_Settings::wordCountForPayload</code> static method . . . . .	15
<b>12</b>	<b>Message RAM</b>	<b>15</b>
<b>13</b>	<b>Sending frames: the <code>tryToSendReturnStatusFD</code> method</b>	<b>16</b>
13.1	Testing a send buffer: the <code>sendBufferNotFullForIndex</code> method . . . . .	17
13.2	Usage example . . . . .	17
<b>14</b>	<b>Retrieving received messages using the <code>receiveFD0</code> method</b>	<b>18</b>
14.1	Driver receive FIFO 0 size . . . . .	19
14.2	The <code>driverReceiveFIFO0Size</code> method . . . . .	20
14.3	The <code>driverReceiveFIFO0Count</code> method . . . . .	20
14.4	The <code>driverReceiveFIFO0PeakCount</code> method . . . . .	20
14.5	The <code>resetDriverReceiveFIFO0PeakCount</code> method . . . . .	20
<b>15</b>	<b>Acceptance filters</b>	<b>20</b>
<b>16</b>	<b>The <code>ACANFD_FeatherM4CAN::begin</code> method reference</b>	<b>21</b>
16.1	The prototype . . . . .	21
16.2	The error code . . . . .	21
16.2.1	The <code>kTxBufferCountGreaterThan32</code> error code . . . . .	21
<b>17</b>	<b><code>ACANFD_FeatherM4CAN_Settings</code> class reference</b>	<b>21</b>
17.1	The <code>ACANFD_FeatherM4CAN_Settings</code> constructor: computation of the CAN bit settings . . . . .	21
17.2	The <code>CANBitSettingConsistency</code> method . . . . .	26
17.3	The <code>actualArbitrationBitRate</code> method . . . . .	27
17.4	The <code>exactArbitrationBitRate</code> method . . . . .	27
17.5	The <code>exactDataBitRate</code> method . . . . .	28
17.6	The <code>ppmFromDesiredArbitrationBitRate</code> method . . . . .	28
17.7	The <code>ppmFromDesiredDataBitRate</code> method . . . . .	28
17.8	The <code>arbitrationSamplePointFromBitStart</code> method . . . . .	28
17.9	The <code>dataSamplePointFromBitStart</code> method . . . . .	28
17.10	Properties of the <code>ACANFD_FeatherM4CAN_Settings</code> class . . . . .	29
17.10.1	The <code>mModuleMode</code> property . . . . .	29
17.10.2	The <code>mEnableRetransmission</code> property . . . . .	29
17.10.3	The <code>mTransceiverDelayCompensation</code> property . . . . .	30
<b>18</b>	<b>Other <code>ACANFD_FeatherM4CAN</code> methods</b>	<b>30</b>
18.1	The <code>errorCounters</code> method . . . . .	30

---

## 1 Versions

Version	Date	Comment
1.0.0	March 8, 2022	Initial release.

## 2 Features

The ACANFD\_FeatherM4CAN library is a CANFD (*Controller Area Network with Flexible Data*) Controller driver for the *Adafruit Feather M4 CAN*<sup>1</sup> board running Arduino. It handles CANFD frames.

This library is compatible with other ACAN libraries.

It has been designed to make it easy to start and to be easily configurable:

- handles the CAN0 and CAN1 CANFD modules;
- default configuration sends and receives any frame – no default filter to provide;
- efficient built-in CAN bit settings computation from arbitration and data bit rates;
- user can fully define its own CAN bit setting values;
- driver and controller transmit buffer sizes are customisable;
- driver and controller receive buffer size is customisable;
- overflow of the driver receive buffer is detectable;
- the message RAM allocation is customizable and the driver checks no overflow occurs;
- *internal loop back*, *external loop back* controller modes are selectable.

## 3 CAN Interfaces

The Adafruit Feather M4 CAN board contains a ATSAME51J19 that implements two CANFD modules: CAN0 and CAN1.

### 3.1 CAN0

The microcontroller CAN0 pins are available on the board connector: D12 is CAN0\_TX, D13 is CAN0\_RX (see [figure 1](#)). For connecting to a CAN bus, you should add a CANFD transceiver. Note D13 is also connected to builtin red led.

---

<sup>1</sup><https://www.adafruit.com/product/4759>




Figure 1 – CAN0 pins

### 3.2 CAN1

The microcontroller CAN1 pins are not available on the board connector, but CANH and CANL pins (see [figure 2](#)). The board includes a 3V-logic compatible transceiver<sup>2</sup>. Note the library handles two additional signals: PIN\_CAN\_STANDBY is configured as low digital output (turning off transceiver's STANDBY mode), and pin 4 is configured as high digital output (turning on transceiver's power).





Figure 2 – CAN1 pins

<sup>2</sup><https://learn.adafruit.com/adafruit-feather-m4-can-express/pinouts>

## 4 Data flow

The [figure 3](#) illustrate default message flow of sending and receiving CANFD messages for CAN0 and CAN1 modules.



**Figure 3** – Message flow in ACANFD\_FeatherM4CAN driver and CANi module, default configuration

**Sending messages.** The ACANFD\_FeatherM4CAN driver defines a *driver transmit FIFO* (default size: 20 messages), and configures the module with a *hardware transmit FIFO* with a size of 24 messages, and 8 individual `TxBuffers` whose capacity is one message.

A message is defined by an instance of the `CANFDMessage` or `CANMessage` class. For sending a message, user code calls the `tryToSendReturnStatusFD` method – see [section 13 page 16](#) for details, and the `idx` property of the sent message should be:

- 0 (default value), for sending via *driver transmit FIFO* and *hardware transmit FIFO*;
- 1, for sending via `TxBuffers0`;
- ...
- 8, for sending via `TxBuffers7`.

---

If the `idx` property is greater than 8, the message is lost.

You can call the `sendBufferNotFullForIndex` method ([section 13.1 page 17](#)) for testing if a send buffer is not full.

**Receiving messages.** The *CAN Protocol Engine* transmits all correct frames to the *reception filters*. By default, they are configured as pass-all, see [section 15 page 20](#) for configuring them. Messages that pass the filters are stored in the *Hardware Reception FIFO0*; its size is 64 messages by default. The interrupt service routine transfers the messages from this FIFO0 to the *Driver Receive FIFO0*. The size of the *Driver Receive Buffer* is 10 by default – see [section 14.1 page 19](#) for changing the default value. Two user methods are available:

- the `availableFD0` method returns `false` if the *Driver Receive FIFO0* is empty, and `true` otherwise;
- the `receiveFD0` method retrieves messages from the *Driver Receive FIFO0* – see [section 14 page 18](#).

## 5 A simple example: LoopBackDemoCANFD\_CAN1

The `LoopBackDemoCANFD_CAN1` sketch is a sample code for introducing the `ACANFD_FeatherM4CAN` library. It demonstrates how to configure the library, to send a CANFD message, and to receive a CANFD message.

Note: this code runs without any CAN connection, the CAN1 module is configured in `EXTERNAL_LOOP_BACK` mode (see [section 17.10.1 page 29](#)); the CAN1 module receives every CANFD frame it sends, and emitted frames can be observed on CANH/CANL pins.

**ACANFD\_FeatherM4CAN configuration.**

```
#define CAN0_MESSAGE_RAM_SIZE (0)
#define CAN1_MESSAGE_RAM_SIZE (1728)

#include <ACANFD_FeatherM4CAN.h>
```

Before including the `ACANFD_FeatherM4CAN` library, you should define the `CAN0_MESSAGE_RAM_SIZE` and the `CAN1_MESSAGE_RAM_SIZE` macro names.

Each CANFD module uses a private *Message RAM* ([section 12 page 15](#)) that is in the first 64 kio of the microcontroller SRAM. Its size depends from the current module configuration, and cannot exceed 4,352 32-bits words (17,408 bytes). Here, `CAN0_MESSAGE_RAM_SIZE` value is 0, meaning that the CAN0 module is not configured; its TxCAN and RxCAN pins can be freely used for an other function. `CAN1_MESSAGE_RAM_SIZE` value is 1728, therefore CAN1 module Message RAM has a capacity of 1,728 32-bit words (6,912 bytes), that corresponds to the default configuration.

Note you should include `<ACANFD_FeatherM4CAN.h>` only once, from the `.ino` source file. From an other C++ file, you should include `<ACANFD_FeatherM4CAN-from-cpp.h>`.

If you include `<ACANFD_FeatherM4CAN.h>` from several files, the `can0` and / or `can1` variables are multiply-defined, therefore you get a link error.

**The setup function.**

```
void setup () {
```

```

//--- Switch on builtin led
pinMode (LED_BUILTIN, OUTPUT) ;
digitalWrite (LED_BUILTIN, HIGH) ;
//--- Start serial
Serial.begin (115200) ;
//--- Wait for serial (blink led at 10 Hz during waiting)
while (!Serial) {
    delay (50) ;
    digitalWrite (LED_BUILTIN, !digitalRead (LED_BUILTIN)) ;
}

```

Builtin led is used for signaling. It blinks led at 10 Hz during until serial monitor is ready.

```

ACANFD_FeatherM4CAN_Settings settings (1000 * 1000, DataBitRateFactor::x2) ;

```

Configuration is a four-step operation. This line is the first step. It instantiates the `settings` object of the `ACANFD_FeatherM4CAN_Settings` class. The constructor has two parameters: the desired CAN arbitration bit rate (here, 1 Mbit/s), and the data bit rate, given by a multiplicative factor of the arbitration bit rate; here, the data bit rate is 1 Mbit/s \* 2 = 2 Mbit/s. It returns a `settings` object fully initialized with CAN bit settings for the desired arbitration and data bit rates, and default values for other configuration properties.

```

settings.mModuleMode = ACANFD_FeatherM4CAN_Settings::EXTERNAL_LOOP_BACK ;

```

This is the second step. You can override the values of the properties of `settings` object. Here, the `mModuleMode` property is set to `EXTERNAL_LOOP_BACK` – its value is `NORMAL_FD` by default. Setting this property enables *external loop back*, that is you can run this demo sketch even if you have no connection to a physical CAN network. The [section 17.10 page 29](#) lists all properties you can override.

```

const uint32_t errorCode = can1.beginFD () ;

```

This is the third step, configuration of the CAN1 driver with `settings` values (for configuring the CAN0 module, use the `can0` variable). The driver is configured for being able to send any (base / extended, data / remote, CAN / CANFD) frame, and to receive all (base / extended, data / remote, CAN / CANFD) frames. If you want to define reception filters, see [section 15 page 20](#).

```

if (errorCode != 0) {
    Serial.print ("Configuration error 0x") ;
    Serial.println (errorCode, HEX) ;
}

```

Last step: the configuration of the can driver returns an error code, stored in the `errorCode` constant. It has the value 0 if all is ok – see [section 16.2 page 21](#).

### The `pseudoRandomValue` function.

This function generates values that are used for generating random CANFD messages.

### The global variables.

```

static const uint32_t PERIOD = 1000 ;
static uint32_t gBlinkDate = PERIOD ;

```

```

static uint32_t gSentCount = 0 ;
static uint32_t gReceiveCount = 0 ;
static CANFDMessage gSentFrame ;
static bool gOk = true ;

```

The gBlinkDate global variable is used for sending a CAN message every second. The gSentCount global variable counts the number of sent messages. The sent message is stored in the gSentFrame variable. While gOk is true, the received message is compared to the sent message. If they are different, gOk is set to false, and no more message is sent. The gReceivedCount global variable counts the number of successfully received messages.

#### The loop function.

```

void loop () {
  if (gBlinkDate <= millis ()) {
    gBlinkDate += PERIOD ;
    digitalWrite (LED_BUILTIN, !digitalRead (LED_BUILTIN)) ;
    if (gOk) {
      ... build random CANFD frame ...
      const uint32_t sendStatus = can1.tryToSendReturnStatusFD (gSentFrame) ;
      if (sendStatus == 0) {
        gSentCount += 1 ;
        Serial.print ("Sent ") ;
        Serial.println (gSentCount) ;
      }else{
        Serial.print ("Sent error 0x") ;
        Serial.println (sendStatus) ;
      }
    }
  }
  //--- Receive frame
  CANFDMessage frame ;
  if (gOk && can1.receiveFD0 (frame)) {
    bool sameFrames = ... compare frame and gSentFrame ... ;
    if (sameFrames) {
      gReceiveCount += 1 ;
      Serial.print ("Received ") ;
      Serial.println (gReceiveCount) ;
    }else{
      gOk = false ;
      ... Print error ...
    }
  }
}

```



---

## 6 The CANMessage class

**Note.** The CANMessage class is declared in the CANMessage.h header file. The class declaration is protected by an include guard that causes the macro GENERIC\_CAN\_MESSAGE\_DEFINED to be defined. The ACAN<sup>3</sup> (version 1.0.3 and above) driver, the ACAN2515<sup>4</sup> driver and the ACAN2517<sup>5</sup> driver contain an identical CANMessage.h file header, enabling using ACAN driver, ACAN2515 driver, ACAN2517 driver and ACAN2517FD driver in a same sketch.

A *CAN message* is an object that contains all CAN 2.0B frame user informations. All properties are initialized by default, and represent a base data frame, with an identifier equal to 0, and without any data. In this library, the CANMessage class is only used by a CANFDMessage constructor (section 7.3 page 10).

```
class CANMessage {
    public : uint32_t id = 0 ; // Frame identifier
    public : bool ext = false ; // false -> standard frame, true -> extended frame
    public : bool rtr = false ; // false -> data frame, true -> remote frame
    public : uint8_t idx = 0 ; // This field is used by the driver
    public : uint8_t len = 0 ; // Length of data (0 ... 8)
    public : union {
        uint64_t data64 ; // Caution: subject to endianness
        int64_t data_s64 ; // Caution: subject to endianness
        uint32_t data32 [2] ; // Caution: subject to endianness
        int32_t data_s32 [2] ; // Caution: subject to endianness
        float dataFloat [2] ; // Caution: subject to endianness
        uint16_t data16 [4] ; // Caution: subject to endianness
        int16_t data_s16 [4] ; // Caution: subject to endianness
        int8_t data_s8 [8] ;
        uint8_t data [8] = {0, 0, 0, 0, 0, 0, 0, 0} ;
    } ;
} ;
```

Note the message datas are defined by an **union**. So message datas can be seen as eight bytes, four 16-bit unsigned integers, two 32-bit, one 64-bit or two 32-bit floats. Be aware that multi-byte integers and floats are subject to endianness (Cortex M4 processors of Teensy 3.x are little-endian).

The idx property is not used in CAN frames, but it is used for selecting the transmit buffer (see section 13 page 16).

## 7 The CANFDMessage class

**Note.** The CANFDMessage class is declared in the CANFDMessage.h header file. The class declaration is protected by an include guard that causes the macro GENERIC\_CANFD\_MESSAGE\_DEFINED to be defined. This

---

<sup>3</sup>The ACAN driver is a CAN driver for FlexCAN modules integrated in the Teensy 3.x microcontrollers, <https://github.com/pierremolinaro/acan>.

<sup>4</sup>The ACAN2515 driver is a CAN driver for the MCP2515 CAN controller, <https://github.com/pierremolinaro/acan2515>.

<sup>5</sup>The ACAN2517 driver is a CAN driver for the MCP2517FD CAN controller in CAN 2.0B mode, <https://github.com/pierremolinaro/acan2517>.

## 7.1 Properties

---

allows an other library to freely include this file without any declaration conflict.

A CANFD message is an object that contains all CANFD frame user informations.

**Example:** The message object describes an extended frame, with identifier equal to 0x123, that contains 12 bytes of data:

```
CANFDMessage message ; // message is fully initialized with default values
message.id = 0x123 ; // Set the message identifier (it is 0 by default)
message.ext = true ; // message is an extended one (it is a base one by default)
message.len = 12 ; // message contains 12 bytes (0 by default)
message.data [0] = 0x12 ; // First data byte is 0x12
...
message.data [11] = 0xCD ; // 11th data byte is 0xCD
```

## 7.1 Properties

```
class CANFDMessage {
    ...
    public : uint32_t id; // Frame identifier
    public : bool ext ; // false -> base frame, true -> extended frame
    public : Type type ;
    public : uint8_t idx ; // Used by the driver
    public : uint8_t len ; // Length of data (0 ... 64)
    public : union {
        uint64_t data64 [ 8] ; // Caution: subject to endianness
        uint32_t data32 [16] ; // Caution: subject to endianness
        uint16_t data16 [32] ; // Caution: subject to endianness
        float dataFloat [16] ; // Caution: subject to endianness
        uint8_t data [64] ;
    } ;
    ...
} ;
```

Note the message datas are defined by an **union**. So message datas can be seen as 64 bytes, 32 x 16-bit unsigned integers, 16 x 32-bit, 8 x 64-bit or 16 x 32-bit floats. Be aware that multi-byte integers are subject to endianness (Cortex M4 processors of Teensy 3.x are little-endian).

## 7.2 The default constructor

All properties are initialized by default, and represent a base data frame, with an identifier equal to 0, and without any data ([table 2](#)).

## 7.3 Constructor from CANMessage

## 7.4 The type property

---

Property	Initial value	Comment
id	0	
ext	false	Base frame
type	CANFD_WITH_BIT_RATE_SWITCH	CANFD frame, with bit rate switch
idx	0	
len	0	No data
data	–	<i>uninitialized</i>

**Table 2** – CANFDMessage default constructor initialization

```
class CANFDMessage {  
    ...  
    CANFDMessage (const CANMessage & inCANMessage) ;  
    ...  
} ;
```

All properties are initialized from the `inCANMessage` ([table 3](#)). Note that only `data64[0]` is initialized from `inCANMessage.data64`.

Property	Initial value
id	<code>inCANMessage.id</code>
ext	<code>inCANMessage.ext</code>
type	<code>inCANMessage.rtr ? CAN_REMOTE : CAN_DATA</code>
idx	<code>inCANMessage.idx</code>
len	<code>inCANMessage.len</code>
<code>data64[0]</code>	<code>inCANMessage.data64</code>

**Table 3** – CANFDMessage constructor CANMessage

## 7.4 The type property

The type property value is an instance of an enumerated type:

```
class CANFDMessage {  
    ...  
    public: typedef enum : uint8_t {  
        CAN_REMOTE,  
        CAN_DATA,  
        CANFD_NO_BIT_RATE_SWITCH,  
        CANFD_WITH_BIT_RATE_SWITCH  
    } Type ;  
    ...  
} ;
```

The type property specifies the frame format, as indicated in the [table 4](#).

## 7.5 The len property

type property	Meaning	Constraint on len
CAN_REMOTE	CAN 2.0B remote frame	0 ... 8
CAN_DATA	CAN 2.0B data frame	0 ... 8
CANFD_NO_BIT_RATE_SWITCH	CANFD frame, no bit rate switch	0 ... 8, 12, 16, 20, 24, 32, 48, 64
CANFD_WITH_BIT_RATE_SWITCH	CANFD frame, bit rate switch	0 ... 8, 12, 16, 20, 24, 32, 48, 64

**Table 4** – CANFDMessage type property

## 7.5 The len property

Note that len property contains the actual length, not its encoding in CANFD frames. So valid values are: 0, 1, ..., 8, 12, 16, 20, 24, 32, 48, 64. Having other values is an error that prevents frame to be sent by the `ACANFD_FeatherM4CAN::tryToSendReturnStatusFD` method. You can use the `pad` method (see [section 7.7 page 12](#)) for padding with `0x00` bytes to the next valid length.

## 7.6 The idx property

The `idx` property is not used in CANFD frames, but it is used for selecting the transmit buffer (see [section 13 page 16](#)).

## 7.7 The pad method

```
void CANFDMessage::pad (void) ;
```

The `CANFDMessage::pad` method appends zero bytes to `datas` for reaching the next valid length. Valid lengths are: 0, 1, ..., 8, 12, 16, 20, 24, 32, 48, 64. If the length is already valid, no padding is performed. For example:

```
CANFDMessage frame ;
frame.length = 21 ; // Not a valid value for sending
frame.pad () ;
// frame.length is 24, frame.data [21], frame.data [22], frame.data [23] are 0
```

## 7.8 The isValid method

```
bool CANFDMessage::isValid (void) const ;
```

Not all settings of `CANFDMessage` instances represent a valid frame. Valid lengths are: 0, 1, ..., 8, 12, 16, 20, 24, 32, 48, 64. For example, there is no CANFD remote frame, so a remote frame should have its length lower than or equal to 8. There is no constraint on extended / base identifier (`ext` property).

The `isValid` returns true if the constraints on the `len` property are checked, as indicated the [table 4 page 12](#), and false otherwise.

---

## 8 Transmit FIFO

The transmit FIFO (see [figure 3 page 5](#)) is composed by:

- the *driver transmit FIFO*, whose size is positive or zero (default 20); you can change the default size by setting the `mDriverTransmitFIFOSize` property of your `settings` object;
- the *hardware transmit FIFO*, whose size is between 1 and 32 (default 24); you can change the default size by setting the `mHardwareTransmitTxFIFOSize` property of your `settings` object.

For sending a message through the *Transmit FIFO*, call the `tryToSendReturnStatusFD` method with a message whose `idx` property is zero:

- if the *controller transmit FIFO* is not full, the message is appended to it, and `tryToSendReturnStatusFD` returns 0;
- otherwise, if the *driver transmit FIFO* is not full, the message is appended to it, and `tryToSendReturnStatusFD` returns 0; the interrupt service routine will transfer messages from *driver transmit FIFO* to the *hardware transmit FIFO* while it is not full;
- otherwise, both FIFOs are full, the message is not stored and `tryToSendReturnStatusFD` returns the `kTransmitBufferOverflow` error.

The transmit FIFO ensures sequentiality of emission.

### 8.1 The `driverTransmitFIFOSize` method

The `driverTransmitFIFOSize` method returns the allocated size of this driver transmit FIFO, that is the value of `settings.mDriverTransmitFIFOSize` when the `begin` method is called.

```
const uint32_t s = can0.driverTransmitFIFOSize () ;
```

### 8.2 The `driverTransmitFIFOCount` method

The `driverTransmitFIFOCount` method returns the current number of messages in the driver transmit FIFO.

```
const uint32_t n = can0.driverTransmitFIFOCount () ;
```

### 8.3 The `driverTransmitFIFOPeakCount` method

The `driverTransmitFIFOPeakCount` method returns the peak value of message count in the driver transmit FIFO

```
const uint32_t max = can0.driverTransmitFIFOPeakCount () ;
```

---

If the transmit FIFO is full when `tryToSendReturnStatusFD` is called, the return value of this call is `kTransmitBufferOverflow`. In such case, the following calls of `driverTransmitBufferPeakCount()` will return `driverTransmitFIFOSize() + 1`.

So, when `driverTransmitFIFOPeakCount()` returns a value lower or equal to `transmitFIFOSize()`, it means that calls to `tryToSendReturnStatusFD` do not provide any overflow of the driver transmit FIFO.

## 9 Transmit buffers (TxBuffer<sub>i</sub>)

You can use `settings.mHardwareDedicatedTxBufferCount` TxBuffers for sending messages. A TxBuffer has a capacity of 1 message. So it is either empty, either full. You can call the `sendBufferNotFullForIndex` method ([section 13.1 page 17](#)) for testing if a TxBuffer is empty or full.

The `settings.mHardwareDedicatedTxBufferCount` property can be set to any integer value between 0 and 32.

## 10 Receive FIFOs

A CAN module contains two receive FIFOs, FIFO0 and FIFO1. **Currently, only FIFO0 is handled, FIFO1 is not configured.**

the receive FIFO<sub>i</sub> ( $0 \leq i \leq 1$ , see [figure 3 page 5](#)) is composed by:

- the *hardware receive FIFO<sub>i</sub>* (in the Message RAM, see [section 12 page 15](#)), whose size is between 0 and 64 (default 64); you can change the default size by setting the `mHardwareRxFIFOiSize` property of your `settings` object;
- the *driver receive FIFO<sub>i</sub>* (in library software), whose size is positive (default 10); you can change the default size by setting the `mDriverReceiveFIFOiSize` property of your `settings` object.

The receive FIFO mechanism ensures sequentiality of reception.

## 11 Payload size

Hardware transmit FIFO, TxBuffers and hardware receive FIFOs objects are stored in the Message RAM, the details of Message RAM usage computation are presented in [section 12 page 15](#). The size of each object depends on the setting applied to the corresponding FIFO or buffer.

By default, all objects accept frames up to 64 data bytes. The size of each object is 72 bytes. If your application sends and / or receives messages with less than 64 bytes, you can reduce Message RAM size by setting the payload properties of `ACANFD_FeatherM4CAN_Settings` class, as described in [table 5](#). The type of these properties is the `ACANFD_FeatherM4CAN_Settings::Payload` enumeration type, and defines 8 values ([table 6](#)).

### 11.1 The ACANFD\_FeatherM4CAN\_Settings::wordCountForPayload static method

Object Size specification	Default value	Applies to
mHardwareTransmitBufferPayload	PAYLOAD_64_BYTES	Hardware transmit FIFO, TxBuffers
mHardwareRxFIFO0Payload	PAYLOAD_64_BYTES	Hardware receive FIFO 0

**Table 5** – Payload properties of ACANFD\_FeatherM4CAN\_Settings class

Object Size specification	Handles frames up to	Object Size
ACANFD_FeatherM4CAN_Settings::PAYLOAD_8_BYTES	8 bytes	4 words = 16 bytes
ACANFD_FeatherM4CAN_Settings::PAYLOAD_12_BYTES	12 bytes	5 words = 20 bytes
ACANFD_FeatherM4CAN_Settings::PAYLOAD_16_BYTES	16 bytes	6 words = 24 bytes
ACANFD_FeatherM4CAN_Settings::PAYLOAD_20_BYTES	20 bytes	7 words = 28 bytes
ACANFD_FeatherM4CAN_Settings::PAYLOAD_24_BYTES	24 bytes	8 words = 32 bytes
ACANFD_FeatherM4CAN_Settings::PAYLOAD_32_BYTES	32 bytes	10 words = 40 bytes
ACANFD_FeatherM4CAN_Settings::PAYLOAD_48_BYTES	48 bytes	14 words = 56 bytes
ACANFD_FeatherM4CAN_Settings::PAYLOAD_64_BYTES	64 bytes	18 words = 72 bytes

**Table 6** – ACANFD\_FeatherM4CAN\_Settings object size from payload size specification

### 11.1 The ACANFD\_FeatherM4CAN\_Settings::wordCountForPayload static method

```
uint32_t ACANFD_FeatherM4CAN_Settings::wordCountForPayload (const Payload inPayload);
```

This static method returns the object word size for a given payload specification, following [table 6](#).

## 12 Message RAM

Each CAN module of the ATSAME51G19A uses a *Message RAM* for storing TxBuffers, hardware transmit FIFO, hardware receives FIFO, and reception filters.

The two Message RAM have a width of 32 bits and are part of ATSAME51G19A SRAM, and they should be located in the first 64 kio (0x2000'0000 – 0x2000'FFFF). Their size is less than 4352 words (17,408 bytes).

A message RAM contains<sup>6</sup>:

- standard filters (0-128 elements, 0-128 words);
- extended filters (0-64 elements, 0-128 words);
- receive FIFO 0 (0-64 elements, 0-1152 words);
- receive FIFO 1 (0-64 elements, 0-1152 words);
- Rx Buffers (0-64 elements, 0-1152 words);
- Tx Event FIFO (0-32 elements, 0-64 words);
- Tx Buffers (0-32 elements, 0-576 words);

<sup>6</sup>See DS60001507G, section 39.9.1 page 1177.

---

So its size cannot exceed 4352 words (17,408 bytes).

The current release of this library allows to define only the following elements:

- receive FIFO 0 (0-64 elements, 0-1152 words);
- Tx Buffers (0-32 elements, 0-576 words);

Its size is therefore actually limited to 1,728 words (6,912 bytes).

There are five properties of `ACANFD_FeatherM4CAN_Settings` class that affect the actual message RAM size:

- the `mHardwareRxFIFO0Size` property sets the hardware receive FIFO 0 element count (0-64);
- the `mHardwareRxFIFO0Payload` property sets the size of the hardware receive FIFO 0 element ([table 6](#));
- the `mHardwareTransmitTxFIFOSize` property sets the hardware transmit FIFO element count (0-32);
- the `mHardwareDedicacedTxBufferCount` property set the number of dedicaced TxBuffers (0-32);
- the `mHardwareTransmitBufferPayload` property sets the size of the TxBuffers and hardware transmit FIFO element ([table 6](#)).

The `ACANFD_FeatherM4CAN::messageRamRequiredSize` method returns the required word size.

The `ACANFD_FeatherM4CAN::begin` method checks the message RAM allocated size is greater or equal to the required size. Otherwise, it raises the error code `kMessageRamTooSmall`. It checks also the message RAM is in the first 64 kio of the SRAM. Otherwise, it raises the error code `kMessageRamNotInFirst64kio`.

## 13 Sending frames: the `tryToSendReturnStatusFD` method

The `ACANFD_FeatherM4CAN::tryToSendReturnStatusFD` method sends CAN 2.0B and CANFD frames:

```
uint32_t ACANFD_FeatherM4CAN::tryToSendReturnStatusFD (const CANFDMessage & inMessage);
```

You call the `tryToSendReturnStatusFD` method for sending a message in the CAN network. Note this function returns before the message is actually sent; this function only adds the message to a transmit buffer. It returns:

- `kInvalidMessage` (value: 1) if the message is not valid (see [section 7.8 page 12](#));
- `kTransmitBufferIndexTooLarge` (value: 2) if the `idx` property value does not specify a valid transmit buffer (see below);
- `kTransmitBufferOverflow` (value: 3) if the transmit buffer specified by the `idx` property value is full;
- 0 (no error) if the message has been successfully added to the transmit buffer specified by the `idx` property value.



### 13.1 Testing a send buffer: the `sendBufferNotFullForIndex` method

---

The `idx` property of the message specifies the transmit buffer:

- 0 for the transmit FIFO ([section 8 page 13](#));
- 1 ... `settings.mHardwareDedicacedTxBufferCount` for a dedicaced TxBuffer ([section 9 page 14](#)).

The type property of `inMessage` specifies how the frame is sent:

- `CAN_REMOTE`, the frame is sent in the CAN 2.0B remote frame format;
- `CAN_DATA`, the frame is sent in the CAN 2.0B data frame format;
- `CANFD_NO_BIT_RATE_SWITCH`, the frame is sent in CANFD format at arbitration bit rate, regardless of the `ACANFD_FeatherM4CAN_Settings::DATA_BITRATE_xn` setting;
- `CANFD_WITH_BIT_RATE_SWITCH`, with the `ACANFD_FeatherM4CAN_Settings::DATA_BITRATE_x1` setting, the frame is sent in CANFD format at arbitration bit rate, and otherwise in CANFD format with bit rate switch.

### 13.1 Testing a send buffer: the `sendBufferNotFullForIndex` method

```
bool ACANFD_FeatherM4CAN::sendBufferNotFullForIndex (const uint32_t inTxBufferIndex);
```

This method returns true if the corresponding transmit buffer is not full, and false otherwise ([table 7](#)).

inTxBufferIndex	Operation
0	true if the transmit FIFO is not full, and false otherwise
1 ... <code>settings.mHardwareDedicacedTxBufferCount</code>	true if the TxBuffer <i>i</i> is empty, and false if it is full
> <code>settings.mHardwareDedicacedTxBufferCount</code>	false

**Table 7** – Value returned by the `sendBufferNotFullForIndex` method

### 13.2 Usage example

A way is to use a global variable to note if the message has been successfully transmitted to driver transmit buffer. For example, for sending a message every 2 seconds:

```
static uint32_t gSendDate = 0 ;

void loop () {
  if (gSendDate < millis ()) {
    CANFDMessage message ;
    // Initialize message properties
    const uint32_t sendStatus = can0.tryToSendReturnStatusFD (message) ;
    if (sendStatus == 0) {
      gSendDate += 2000 ;
    }
  }
}
```

---

An other hint to use a global boolean variable as a flag that remains true while the message has not been sent.

```
static bool gSendMessage = false ;

void loop () {
    ...
    if (frame_should_be_sent) {
        gSendMessage = true ;
    }
    ...
    if (gSendMessage) {
        CANMessage message ;
        // Initialize message properties
        const uint32_t sendStatus = can0.tryToSendReturnStatusFD (message) ;
        if (sendStatus == 0) {
            gSendMessage = false ;
        }
    }
    ...
}
```

## 14 Retrieving received messages using the receiveFD0 method

```
bool ACANFD_FeatherM4CAN::receiveFD0 (CANFDMessage & outMessage) ;
```

If the receive FIFO 0 is not empty, the oldest message is removed, assigned to outMessage, and the method returns true. If the receive FIFO 0 is empty, the method returns false.

This is a basic example:

```
void loop () {
    CANFDMessage message ;
    if (can0.receiveFD0 (message)) {
        // Handle received message
    }
    ...
}
```

The receive method:

- returns false if the driver receive buffer is empty, message argument is not modified;
- returns true if a message has been removed from the driver receive buffer, and the message argument is assigned.

The type property contains the received frame format:

- CAN\_REMOTE, the received frame is a CAN 2.0B remote frame;
- CAN\_DATA, the received frame is a CAN 2.0B data frame;
- CANFD\_NO\_BIT\_RATE\_SWITCH, the frame received frame is a CANFD frame, received at at arbitration bit rate;
- CANFD\_WITH\_BIT\_RATE\_SWITCH, the frame received frame is a CANFD frame, received with bit rate switch.

You need to manually dispatch the received messages. If you did not provide any receive filter, you should check the type property (remote or data frame?), the ext bit (base or extended frame), and the id (identifier value). The following snippet dispatches three messages:

```
void loop () {
    CANFDMessage message ;
    if (can0.receive (message)) {
        if (!message.rtr && message.ext && (message.id == 0x123456)) {
            handle_myMessage_0 (message) ; // Extended data frame, id is 0x123456
        } else if (!message.rtr && !message.ext && (message.id == 0x234)) {
            handle_myMessage_1 (message) ; // Base data frame, id is 0x234
        } else if (message.rtr && !message.ext && (message.id == 0x542)) {
            handle_myMessage_2 (message) ; // Base remote frame, id is 0x542
        }
    }
    ...
}
```

The handle\_myMessage\_0 function has the following header:

```
void handle_myMessage_0 (const CANFDMessage & inMessage) {
    ...
}
```

So are the header of the handle\_myMessage\_1 and the handle\_myMessage\_2 functions.

## 14.1 Driver receive FIFO 0 size

By default, the driver receive FIFO 0 size is 10. You can change it by setting the mDriverReceiveFIFO0Size property of settings variable before calling the begin method:

```
ACANFD_FeatherM4CAN_Settings settings (125 * 1000,
                                         DataBitRateFactor::x4) ;
settings.mDriverReceiveFIFO0Size = 100 ;
const uint32_t errorCode = can0.begin (settings) ;
...
```

As the size of CANFDMessage class is 72 bytes, the actual size of the driver receive FIFO 0 is the value of settings.mDriverReceiveFIFO0Size \* 72.

### 14.2 The driverReceiveFIFO0Size method

The driverReceiveFIFO0Size method returns the size of the driver FIFO 0, that is the value of the mDriverReceiveFIFO0 property of settings variable when the begin method is called.

```
const uint32_t s = can0.driverReceiveFIFO0Size ();
```

### 14.3 The driverReceiveFIFO0Count method

The driverReceiveFIFO0Count method returns the current number of messages in the driver receive FIFO 0.

```
const uint32_t n = can0.driverReceiveFIFO0Count ();
```

### 14.4 The driverReceiveFIFO0PeakCount method

The driverReceiveFIFO0PeakCount method returns the peak value of message count in the driver receive FIFO 0.

```
const uint32_t max = can0.driverReceiveFIFO0PeakCount ();
```

If an overflow occurs, further calls of can0.receiveBufferPeakCount () return can0.receiveBufferSize ()+1.

### 14.5 The resetDriverReceiveFIFO0PeakCount method

The resetDriverReceiveFIFO0PeakCount method assign the current count to the peak value.

```
can0.resetDriverReceiveFIFO0PeakCount ();
```

## 15 Acceptance filters

In the current release of the library, the only receive filters that can be defined are for remote frames:

- setting the mDiscardReceivedStandardRemoteFrames property of the ACANFD\_FeatherM4CAN\_Settings class discards every received remote frame with a standard identifier;
- setting the mDiscardReceivedExtendedRemoteFrames property of the ACANFD\_FeatherM4CAN\_Settings class discards every received remote frame with an extended identifier.

By default, theses properties are set to true, meaning all remote frames are received.

---

## 16 The ACANFD\_FeatherM4CAN::begin method reference

### 16.1 The prototype

```
uint32_t ACANFD_FeatherM4CAN::begin (const ACANFD_FeatherM4CAN_Settings & inSettings) ;
```

This prototype has one argument, a `ACANFD_FeatherM4CAN_Settings` instance that defines the settings. It configures the controller in such a way that all messages are received (*pass-all* filter).

### 16.2 The error code

The `ACANFD_FeatherM4CAN::begin` method returns an error code. The value `0` denotes no error. Otherwise, you consider every bit as an error flag, as described in [table 8](#). An error code could report several errors. The `ACANFD_FeatherM4CAN` class defines static constants for naming errors. Bits 0 to 16 denote a bit configuration error, see [table 10 page 27](#).

Bit	Code	Static constant Name	Comment
0	0x1	kBitRatePrescalerIsZero	See <a href="#">table 10 page 27</a>
...	...	...	See <a href="#">table 10 page 27</a>
16	0x1_0000	kDataSJWIsGreaterThanPhaseSegment2	See <a href="#">table 10 page 27</a>
20	0x10_0000	kMessageRamTooSmall	See <a href="#">section 12 page 15</a>
21	0x20_0000	kMessageRamNotInFirst64kio	See <a href="#">section 12 page 15</a>
22	0x40_0000	kHardwareRxFIFOSizeGreaterThan64	<code>settings.mHardwareRxFIFOSize &gt; 64</code>
23	0x80_0000	kHardwareTransmitFIFOSizeGreaterThan32	<code>settings.mHardwareTransmitTxFIFOSize &gt; 32</code>
24	0x100_0000	kDedicacedTransmitTxBufferCountGreaterThan32	<code>settings.mHardwareDedicacedTxBufferCount &gt; 32</code>
25	0x200_0000	kTxBufferCountGreaterThan32	See <a href="#">section 16.2.1 page 21</a>

**Table 8** – The `ACANFD_FeatherM4CAN::begin` method error code bits

#### 16.2.1 The `kTxBufferCountGreaterThan32` error code

There are 32 available `TxBuffers`, for hardware transmit FIFO and dedicaced `TxBuffers`. Therefore, the sum of `settings.mHardwareDedicacedTxBufferCount` and `settings.mHardwareTransmitTxFIFOSize` should be lower or equal to 32.

## 17 ACANFD\_FeatherM4CAN\_Settings class reference

**Note.** The `ACANFD_FeatherM4CAN_Settings` class is not Arduino specific. You can compile it on your desktop computer with your favorite C++ compiler.

### 17.1 The `ACANFD_FeatherM4CAN_Settings` constructor: computation of the CAN bit settings

```
void setup () {
  ACANFD_FeatherM4CAN_Settings::
  ACANFD_FeatherM4CAN_Settings (const uint32_t inDesiredArbitrationBitRate,
                                const DataBitRateFactor inDataBitRateFactor,
                                const uint32_t inTolerancePPM = 1000) ;
}
```

The constructor of the ACANFD\_FeatherM4CAN\_Settings has two mandatory arguments: the desired arbitration bit rate, and the data bit rate factor. It tries to compute the CAN bit settings for these bit rates. If it succeeds, the constructed object has its `mArbitrationBitRateClosedToDesiredRate` property set to `true`, otherwise it is set to `false`. For example, for an 1 Mbit/s arbitration bit rate and an 8 Mbit/s data bit rate:

```
void setup () {
  // Arbitration bit rate: 1 Mbit/s, data bit rate: 8 Mbit/s
  ACANFD_FeatherM4CAN_Settings settings (1000 * 1000, DataBitRateFactor::x8) ;
  // Here, settings.mArbitrationBitRateClosedToDesiredRate is true
  ...
}
```

Note the data bit rate is not defined by its frequency, but by its multiplicative factor from arbitration bit rate. If you want a single bit rate, use `DataBitRateFactor::x1` as data bit rate factor.

There are 313 exact arbitration / data bit rate combinations ([table 9 page 23](#)).

But this does not mean there is no possibility to get such data bit rates factors. For example, we can have a data bit rate of 4 Mbit/s, and an arbitration bit rate of  $4/7$  Mbit/s = 571 428 kbit/s:

```
void setup () {
  ...
  ACANFD_FeatherM4CAN_Settings settings (571428, DataBitRateFactor::x7) ;
  Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
  Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 1 (--> is true)
  Serial.print ("Actual Arbitration Bit Rate: ") ;
  Serial.println (settings.actualArbitrationBitRate ()) ; // 571428 bit/s
  Serial.print ("distance: ") ;
  Serial.println (settings.ppmFromDesiredArbitrationBitRate ()) ; // 1 ppm= 0,0001 %
  Serial.print ("Actual Data Bit Rate: ") ;
  Serial.println (settings.actualDataBitRate ()) ; // 4 Mbit/s
  ...
}
```

Due to integer computations, and the distance from desired arbitration bit rate is 1 ppm. "ppm" stands for "part-per-million", and  $1 \text{ ppm} = 10^{-6}$ . In other words,  $10,000 \text{ ppm} = 1\%$ .

By default, a desired bit rate is accepted if the distance from the computed actual bit rate is lower or equal to  $1,000 \text{ ppm} = 0.1\%$ . You can change this default value by adding your own value as third argument of ACANFD\_FeatherM4CAN\_Settings constructor. For example, with an arbitration bit rate equal to 727 kbit/s:

```
void setup () {
  ...
  ACANFD_FeatherM4CAN_Settings settings (727 * 1000, DataBitRateFactor::x1, 100) ; // 100 ppm
  Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
}
```

## 17.1 The ACANFD\_FeatherM4CAN\_Settings constructor: computation of the CAN bit settings

Arbitration Bit Rate	Valid Data Rate factors	Arbitration Bit Rate	Valid Data Rate factors
5 000	x8 x10	6 000	x8 x10
6 250	x5 x6 x8 x10	6 400	x10
7 500	x5 x8 x10	7 680	x10
8 000	x5 x6 x8 x10	9 375	x4 x5 x8 x10
9 600	x5 x8 x10	10 000	x4 x5 x6 x8 x10
12 000	x4 x5 x8 x10	12 500	x3 x4 x5 x6 x8 x10
12 800	x5 x6 x10	15 000	x4 x5 x8 x10
15 360	x5	15 625	x2 x3 x4 x6 x8
16 000	x3 x4 x5 x6 x8 x10	18 750	x2 x4 x5 x8 x10
19 200	x4 x5 x10	20 000	x2 x3 x4 x5 x6 x8 x10
24 000	x2 x4 x5 x8 x10	25 000	x2 x3 x4 x5 x6 x8 x10
25 600	x3 x5	30 000	x2 x4 x5 x8 x10
31 250	x1 x2 x3 x4 x6 x8	32 000	x2 x3 x4 x5 x6 x10
37 500	x1 x2 x4 x5 x8 x10	38 400	x2 x5 x10
40 000	x1 x2 x3 x4 x5 x6 x8 x10	46 875	x1 x2 x4 x8
48 000	x1 x2 x4 x5 x8 x10	50 000	x1 x2 x3 x4 x5 x6 x8 x10
60 000	x1 x2 x4 x5 x8 x10	62 500	x1 x2 x3 x4 x6 x8
64 000	x1 x2 x3 x5 x6 x10	75 000	x1 x2 x4 x5 x8 x10
76 800	x1 x5	80 000	x1 x2 x3 x4 x5 x6 x8 x10
93 750	x1 x2 x4 x8	96 000	x1 x2 x4 x5 x10
100 000	x1 x2 x3 x4 x5 x6 x8 x10	120 000	x1 x2 x4 x5 x8 x10
125 000	x1 x2 x3 x4 x6 x8	128 000	x1 x3 x5
150 000	x1 x2 x4 x5 x8 x10	160 000	x1 x2 x3 x4 x5 x6 x10
187 500	x1 x2 x4 x8	192 000	x1 x2 x5 x10
200 000	x1 x2 x3 x4 x5 x6 x8 x10	240 000	x1 x2 x4 x5 x8 x10
250 000	x1 x2 x3 x4 x6 x8	300 000	x1 x2 x4 x5 x8 x10
320 000	x1 x2 x3 x5 x6 x10	375 000	x1 x2 x4 x8
384 000	x1 x5	400 000	x1 x2 x3 x4 x5 x6 x8 x10
480 000	x1 x2 x4 x5 x10	500 000	x1 x2 x3 x4 x6 x8
600 000	x1 x2 x4 x5 x8 x10	640 000	x1 x3 x5
750 000	x1 x2 x4 x8	800 000	x1 x2 x3 x4 x5 x6 x10
960 000	x1 x2 x5 x10	1 000 000	x1 x2 x3 x4 x6 x8

**Table 9** – The 313 exact bit rates

```

Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 0 (--> is false)
Serial.print ("actual arbitration bit rate: ") ;
Serial.println (settings.actualArbitrationBitRate ()) ; // 727272 bit/s
Serial.print ("distance: ") ;
Serial.println (settings.ppmFromDesiredArbitrationBitRate ()) ; // 375 ppm
...
}

```

The third argument does not change the CAN bit computation, it only changes the acceptance test for setting the `mArbitrationBitRateClosedToDesiredRate` property. For example, you can specify that you want the computed actual bit to be exactly the desired bit rate:

```

void setup () {
    ...
    ACANFD_FeatherM4CAN_Settings settings (500 * 1000,
                                             DataBitRateFactor::x1,
                                             0) ; // Max distance is 0 ppm
    Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
    Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 1 (--> is true)
}

```

## 17.1 The ACANFD\_FeatherM4CAN\_Settings constructor: computation of the CAN bit settings

```
Serial.print ("actual arbitration bit rate: ") ;
Serial.println (settings.actualArbitrationBitRate ()) ; // 500,000 bit/s
Serial.print ("distance: ") ;
Serial.println (settings.ppmFromDesiredArbitrationBitRate ()) ; // 0 ppm
...
}
```

In any way, the bit rate computation always gives a consistent result, resulting an actual arbitration / data bit rates closest from the desired bit rate. For example, we query a 423 kbit/s arbitration bit rate, and a 423 kbit/s \* 3 = 1 269 kbit/s data bit rate:

```
void setup () {
  ...
  ACANFD_FeatherM4CAN_Settings settings (423 * 1000, DataBitRateFactor::x3) ;
  Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
  Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 0 (--> is false)
  Serial.print ("Actual Arbitration Bit Rate: ") ;
  Serial.println (settings.actualArbitrationBitRate ()) ; // 421 052 bit/s
  Serial.print ("Actual Data Bit Rate: ") ;
  Serial.println (settings.actualDataBitRate ()) ; // 1 263 157 bit/s
  Serial.print ("distance: ") ;
  Serial.println (settings.ppmFromDesiredArbitrationBitRate ()) ; // 4 603 ppm
  ...
}
```

The resulting bit rates settings are far from the desired values, the CAN bit decomposition is consistent. You can get its details:

```
void setup () {
  ...
  ACANFD_FeatherM4CAN_Settings settings (423 * 1000, DataBitRateFactor::x3) ;
  Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
  Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 0 (--> is false)
  Serial.print ("Actual Arbitration Bit Rate: ") ;
  Serial.println (settings.actualArbitrationBitRate ()) ; // 421 052 bit/s
  Serial.print ("Actual Data Bit Rate: ") ;
  Serial.println (settings.actualDataBitRate ()) ; // 1 263 157 bit/s
  Serial.print ("distance: ") ;
  Serial.println (settings.ppmFromDesiredArbitrationBitRate ()) ; // 4 603 ppm
  Serial.print ("Bit rate prescaler: ") ;
  Serial.println (settings.mBitRatePrescaler) ; // BRP = 1
  Serial.print ("Arbitration Phase segment 1: ") ;
  Serial.println (settings.mArbitrationPhaseSegment1) ; // PS1 = 22
  Serial.print ("Arbitration Phase segment 2: ") ;
  Serial.println (settings.mArbitrationPhaseSegment2) ; // PS2 = 10
  Serial.print ("Arbitration Resynchronization Jump Width: ") ;
  Serial.println (settings.mArbitrationSJW) ; // SJW = 10
  Serial.print ("Arbitration Sample Point: ") ;
  Serial.println (settings.arbitrationSamplePointFromBitStart ()) ; // 69, meaning 69%
```



```
Serial.print ("Data Phase segment 1: ") ;
Serial.println (settings.mDataPhaseSegment1) ; // PS1 = 22
Serial.print ("Data Phase segment 2: ") ;
Serial.println (settings.mDataPhaseSegment2) ; // PS2 = 10
Serial.print ("Data Resynchronization Jump Width: ") ;
Serial.println (settings.mDataSJW) ; // SJW = 10
Serial.print ("Data Sample Point: ") ;
Serial.println (settings.dataSamplePointFromBitStart ()) ; // 69, meaning 59%
Serial.print ("Consistency: ") ;
Serial.println (settings.CANBitSettingConsistency ()) ; // 0, meaning Ok
...
}
```

The `samplePointFromBitStart` method returns sample point, expressed in per-cent of the bit duration from the beginning of the bit.

Note the computation may calculate a bit decomposition too far from the desired bit rate, but it is always consistent. You can check this by calling the `CANBitSettingConsistency` method.

You can change the property values for adapting to the particularities of your CAN network propagation time. By example, you can increment the `mArbitrationPhaseSegment1` property value, and decrement the `mArbitrationPhaseSegment2` property value in order to sample the CAN Rx pin later.

```
void setup () {
    ...
    ACANFD_FeatherM4CAN_Settings settings (500 * 1000, DataBitRateFactor::x1) ;
    Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
    Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 1 (--> is true)
    settings.mArbitrationPhaseSegment1 -= 4 ; // 32 -> 28: safe, 1 <= PS1 <= 256
    settings.mArbitrationPhaseSegment2 += 4 ; // 15 -> 19: safe, 1 <= PS2 <= 128
    settings.mArbitrationSJW += 4 ; // 15 -> 19: safe, 1 <= SJW <= PS2
    Serial.print ("Sample Point: ") ;
    Serial.println (settings.samplePointFromBitStart ()) ; // 58, meaning 58%
    Serial.print ("actual arbitration bit rate: ") ;
    Serial.println (settings.actualArbitrationBitRate ()) ; // 500000: ok, no change
    Serial.print ("Consistency: ") ;
    Serial.println (settings.CANBitSettingConsistency ()) ; // 0, meaning Ok
    ...
}
```

Be aware to always respect CAN bit timing consistency! The ATSAME51G19A constraints are:

$$\begin{aligned}
 1 &\leq \text{mBitRatePrescaler} \leq 32 \\
 1 &\leq \text{mArbitrationPhaseSegment1} \leq 256 \\
 2 &\leq \text{mArbitrationPhaseSegment2} \leq 128 \\
 1 &\leq \text{mArbitrationSJW} \leq \text{mArbitrationPhaseSegment2} \\
 1 &\leq \text{mDataPhaseSegment1} \leq 32 \\
 2 &\leq \text{mDataPhaseSegment2} \leq 16 \\
 1 &\leq \text{mDataSJW} \leq \text{mDataPhaseSegment2}
 \end{aligned}$$

Miicrochips recommends using the same bit rate prescaler for arbitration and data bit rates.

Resulting actual bit rates are given by (SYSCLK = 48 MHz):

$$\begin{aligned}
 \text{Actual Arbitration Bit Rate} &= \frac{\text{SYSCLK}}{\text{mBitRatePrescaler} \cdot (1 + \text{mArbitrationPhaseSegment1} + \text{mArbitrationPhaseSegment2})} \\
 \text{Actual Data Bit Rate} &= \frac{\text{SYSCLK}}{\text{mBitRatePrescaler} \cdot (1 + \text{mDataPhaseSegment1} + \text{mDataPhaseSegment2})}
 \end{aligned}$$

And the sampling point (in per-cent unit) are given by:

$$\begin{aligned}
 \text{Arbitration Sampling Point} &= 100 \cdot \frac{1 + \text{mArbitrationPhaseSegment1}}{1 + \text{mArbitrationPhaseSegment1} + \text{mArbitrationPhaseSegment2}} \\
 \text{Data Sampling Point} &= 100 \cdot \frac{1 + \text{mDataPhaseSegment1}}{1 + \text{mDataPhaseSegment1} + \text{mDataPhaseSegment2}}
 \end{aligned}$$

## 17.2 The CANBitSettingConsistency method

This method checks the CAN bit decomposition (given by mBitRatePrescaler, mArbitrationPhaseSegment1, mArbitrationPhaseSegment2, mArbitrationSJW, mDataPhaseSegment1, mDataPhaseSegment2, mDataSJW property values) is consistent.

```

void setup () {
    ...
    ACANFD_FeatherM4CAN_Settings settings (500 * 1000, DataBitRateFactor::x2) ;
    Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
    Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 1 (--> is true)
    settings.mDataPhaseSegment1 = 0 ; // Error, mDataPhaseSegment1 should be >= 1 (and <= 32)
    Serial.print ("Consistency: 0x") ;
    Serial.println (settings.CANBitSettingConsistency (), HEX) ; // != 0, meaning error
    ...
}

```

The CANBitSettingConsistency method returns 0 if CAN bit decomposition is consistent. Otherwise, the returned value is a bit field that can report several errors – see [table 10](#).

### 17.3 The actualArbitrationBitRate method

The ACANFD\_FeatherM4CAN\_Settings class defines static constant properties that can be used as mask error. For example:

```
public: static const uint32_t kBitRatePrescalerIsZero = 1 << 0 ;
```

Bit	Code	Error Name	Error
0	0x1	kBitRatePrescalerIsZero	mBitRatePrescaler == 0
1	0x2	kBitRatePrescalerIsGreaterThan32	mBitRatePrescaler > 32
2	0x4	kArbitrationPhaseSegment1IsZero	mArbitrationPhaseSegment1 == 0
3	0x8	kArbitrationPhaseSegment1IsGreaterThan256	mArbitrationPhaseSegment1 > 256
4	0x10	kArbitrationPhaseSegment2IsLowerThan2	mArbitrationPhaseSegment2 < 2
5	0x20	kArbitrationPhaseSegment2IsGreaterThan128	mArbitrationPhaseSegment2 > 128
6	0x40	kArbitrationSJWIsZero	mArbitrationSJW == 0
7	0x80	kArbitrationSJWIsGreaterThan128	mArbitrationSJW > 128
8	0x100	kArbitrationSJWIsGreaterThanPhaseSegment2	mArbitrationSJW > mArbitrationPhaseSegment2
9	0x200	kArbitrationPhaseSegment1Is1AndTripleSampling	(mArbitrationPhaseSegment1 == 1) and triple sampling
10	0x400	kDataPhaseSegment1IsZero	mDataPhaseSegment1 == 0
11	0x800	kDataPhaseSegment1IsGreaterThan32	mDataPhaseSegment1 > 32
12	0x1000	kDataPhaseSegment2IsLowerThan2	mDataPhaseSegment2 < 2
13	0x2000	kDataPhaseSegment2IsGreaterThan16	mDataPhaseSegment2 > 16
14	0x4000	kDataSJWIsZero	mDataSJW == 0
15	0x8000	kDataSJWIsGreaterThan16	mDataSJW > 16
16	0x1_0000	kDataSJWIsGreaterThanPhaseSegment2	mDataSJW > mDataPhaseSegment2

**Table 10** – The ACANFD\_FeatherM4CAN\_Settings::CANBitSettingConsistency method error codes

### 17.3 The actualArbitrationBitRate method

The actualArbitrationBitRate method returns the actual bit computed from mBitRatePrescaler, mPropagationSegment, mArbitrationPhaseSegment1, mArbitrationPhaseSegment2, mArbitrationSJW property values.

```
void setup () {
    ...
    ACANFD_FeatherM4CAN_Settings settings (440 * 1000, DataBitRateFactor::x1) ;
    Serial.print ("mArbitrationBitRateClosedToDesiredRate: ") ;
    Serial.println (settings.mArbitrationBitRateClosedToDesiredRate) ; // 0 (--> is false)
    Serial.print ("actual arbitration bit rate: ") ;
    Serial.println (settings.actualArbitrationBitRate ()) ; // 444,444 bit/s
    ...
}
```

**Note.** If CAN bit settings are not consistent (see [section 17.2 page 26](#)), the returned value is irrelevant.

### 17.4 The exactArbitrationBitRate method

```
bool ACANFD_FeatherM4CAN_Settings::exactArbitrationBitRate (void) const ;
```

## 17.5 The exactDataBitRate method

---

The `exactArbitrationBitRate` method returns `true` if the actual arbitration bit rate is equal to the desired arbitration bit rate, and `false` otherwise.

**Note.** If CAN bit settings are not consistent (see [section 17.2 page 26](#)), the returned value is irrelevant.

## 17.5 The exactDataBitRate method

```
bool ACANFD_FeatherM4CAN_Settings::exactDataBitRate (void) const ;
```

The `exactDataBitRate` method returns `true` if the actual data bit rate is equal to the desired data bit rate, and `false` otherwise.

**Note.** If CAN bit settings are not consistent (see [section 17.2 page 26](#)), the returned value is irrelevant.

## 17.6 The ppmFromDesiredArbitrationBitRate method

```
uint32_t ACANFD_FeatherM4CAN_Settings::ppmFromDesiredArbitrationBitRate (void) const ;
```

The `ppmFromDesiredArbitrationBitRate` method returns the distance from the actual arbitration bit rate to the desired arbitration bit rate, expressed in part-per-million (ppm):  $1 \text{ ppm} = 10^{-6}$ . In other words,  $10,000 \text{ ppm} = 1\%$ .

**Note.** If CAN bit settings are not consistent (see [section 17.2 page 26](#)), the returned value is irrelevant.

## 17.7 The ppmFromDesiredDataBitRate method

```
uint32_t ACANFD_FeatherM4CAN_Settings::ppmFromDesiredDataBitRate (void) const ;
```

The `ppmFromDesiredDataBitRate` method returns the distance from the actual data bit rate to the desired data bit rate, expressed in part-per-million (ppm):  $1 \text{ ppm} = 10^{-6}$ . In other words,  $10,000 \text{ ppm} = 1\%$ .

**Note.** If CAN bit settings are not consistent (see [section 17.2 page 26](#)), the returned value is irrelevant.

## 17.8 The arbitrationSamplePointFromBitStart method

```
uint32_t ACANFD_FeatherM4CAN_Settings::arbitrationSamplePointFromBitStart (void) const ;
```

The `arbitrationSamplePointFromBitStart` method returns the distance of sample point from the start of the arbitration CAN bit, expressed in part-per-cent (ppc):  $1 \text{ ppc} = 1\% = 10^{-2}$ . It is a good practice to get sample point from 65% to 80%. The bit rate calculator tries to set the sample point at 80%.

**Note.** If CAN bit settings are not consistent (see [section 17.2 page 26](#)), the returned value is irrelevant.

## 17.9 The dataSamplePointFromBitStart method

```
uint32_t ACANFD_FeatherM4CAN_Settings::dataSamplePointFromBitStart (void) const ;
```

The `dataSamplePointFromBitStart` method returns the distance of sample point from the start of the data CAN bit, expressed in part-per-cent (ppc):  $1 \text{ ppc} = 1\% = 10^{-2}$ . It is a good practice to get sample point from 65% to 80%. The bit rate calculator tries to set the sample point at 80%.

**Note.** If CAN bit settings are not consistent (see [section 17.2 page 26](#)), the returned value is irrelevant.

## 17.10 Properties of the ACANFD\_FeatherM4CAN\_Settings class

All properties of the `ACANFD_FeatherM4CAN_Settings` class are declared public and are initialized ([table 11](#)).

Property	Type	Initial value	Comment
<code>mDesiredArbitrationBitRate</code>	<code>uint32_t</code>	Constructor argument	
<code>mDataBitRateFactor</code>	<code>DataBitRateFactor</code>	Constructor argument	
<code>mBitRatePrescaler</code>	<code>uint8_t</code>	32	See <a href="#">section 17.1 page 21</a>
<code>mArbitrationPhaseSegment1</code>	<code>uint16_t</code>	256	See <a href="#">section 17.1 page 21</a>
<code>mArbitrationPhaseSegment2</code>	<code>uint8_t</code>	128	See <a href="#">section 17.1 page 21</a>
<code>mArbitrationSJW</code>	<code>uint8_t</code>	128	See <a href="#">section 17.1 page 21</a>
<code>mDataPhaseSegment1</code>	<code>uint8_t</code>	32	See <a href="#">section 17.1 page 21</a>
<code>mDataPhaseSegment2</code>	<code>uint8_t</code>	16	See <a href="#">section 17.1 page 21</a>
<code>mDataSJW</code>	<code>uint8_t</code>	16	See <a href="#">section 17.1 page 21</a>
<code>mTripleSampling</code>	<code>bool</code>	true	See <a href="#">section 17.1 page 21</a>
<code>mBitSettingOk</code>	<code>bool</code>	true	See <a href="#">section 17.1 page 21</a>
<code>mModuleMode</code>	<code>ModuleMode</code>	<code>NORMAL_FD</code>	See <a href="#">section 17.10.1 page 29</a>
<code>mDriverReceiveFIFO0Size</code>	<code>uint16_t</code>	10	See <a href="#">section 14.1 page 19</a>
<code>mHardwareRxFIFO0Size</code>	<code>uint8_t</code>	64	See <a href="#">section 12 page 15</a>
<code>mHardwareRxFIFO0Payload</code>	<code>Payload</code>	<code>PAYLOAD_64_BYTES</code>	See <a href="#">section 12 page 15</a>
<code>mEnableRetransmission</code>	<code>bool</code>	true	See <a href="#">section 17.10.2 page 29</a>
<code>mDiscardReceivedStandardRemoteFrames</code>	<code>bool</code>	false	See <a href="#">section 15 page 20</a>
<code>mDiscardReceivedExtendedRemoteFrames</code>	<code>bool</code>	false	See <a href="#">section 15 page 20</a>
<code>mTransceiverDelayCompensation</code>	<code>uint8_t</code>	5	See <a href="#">section 17.10.3 page 30</a>
<code>mDriverTransmitFIFOSize</code>	<code>uint8_t</code>	20	See <a href="#">section 8 page 13</a>
<code>mHardwareTransmitTxFIFOSize</code>	<code>uint8_t</code>	24	See <a href="#">section 8 page 13</a>
<code>mHardwareDedicatedTxBufferCount</code>	<code>uint8_t</code>	8	See <a href="#">section 9 page 14</a>
<code>mHardwareTransmitBufferPayload</code>	<code>Payload</code>	<code>PAYLOAD_64_BYTES</code>	See <a href="#">section 11 page 14</a>

**Table 11** – Properties of the `ACANFD_FeatherM4CAN_Settings` class

### 17.10.1 The `mModuleMode` property

This property defines the mode requested at this end of the configuration process: `NORMAL_FD` (default value), `INTERNAL_LOOP_BACK`, `EXTERNAL_LOOP_BACK`.

### 17.10.2 The `mEnableRetransmission` property

By default, a frame is automatically retransmitted if an error occurs during its transmission, or if its transmission is preempted by a higher priority frame. You can turn off this feature by setting the `mEnableRetransmission` to false.

---

### 17.10.3 The `mTransceiverDelayCompensation` property

Setting the *Transmitter Delay Compensation* is required when data bit rate switch is enabled and data phase bit time that is shorter than the transceiver loop delay. The `mTransceiverDelayCompensation` property is by default set to 8 by the `ACANFD_FeatherM4CAN_Settings` constructor.

For more details, see DS60001507G, sections 39.6.2.4, pages 1095 and 1096.

## 18 Other `ACANFD_FeatherM4CAN` methods

### 18.1 The `errorCounters` method

```
uint32_t ACANFD_FeatherM4CAN::errorCounters (void) ;
```

This method returns the transmit / receive error count register value, as described in DS20005688B, REGISTER 3-19 page 41. The `CiTREC` value is zero when there is no error.